
FINISH 
/CLEAR 
C********************************************************* 
C*** 
C*** TUBO INTERRATO IN PRESSIONE IN MATERIALE COMPOSITO 
C*** 
C*** PARAMETRI 
C*** 
/PREP7 
D=800 
SP=9 
L2=4*D 
N1=10 
L1=0.1*D 
N2=20 
NC=12 
RR=5 
E1=0.0025 
E2=0.14 
ANG1=105 
ANG2=15 
SRPT=600 
SRPC=650 
SROT=30 
SROC=90 
SRT=45 
TETA=60 
SPL=0.25 
NL=NINT(SP/SPL) 
*IF,NL,GT,NINT(NL/2)*2,THEN 
      NL=NL+1 
*ENDIF 
PMX=1800*9.81*2.5/1E6 
C*** 
C*** NODI 
C*** 
CSYS,1 
N,1,D/2,-90 
N,NC+1,D/2,90 
FILL,1,NC+1 
NGEN,N1+1,NC+1,1,NC+1,,0,0,L1/N1 
NGEN,N2+1,NC+1,(NC+1)*N1+1,(NC+1)*(N1+1),,0,0,L2/N2,RR 
C*** 
C*** ELEMENTI 
C*** 
ET,1,99 
R,1,NL,0,,,0, 
RMORE 
*DO,IJK,1,NL/2 
RMORE,1,TETA,SPL,1,-TETA,SPL 
*ENDDO 
R,2,NL,0,,,E1 
RMORE 
*DO,IJK,1,NL/2 
RMORE,1,TETA,SPL,1,-TETA,SPL 
*ENDDO 
R,3,NL,0,,,E2 
RMORE 
*DO,IJK,1,NL/2 
RMORE,1,TETA,SPL,1,-TETA,SPL 
*ENDDO 
E,1,NC+2,NC+3,2 
EGEN,NC,1,1 



EGEN,N1+N2,NC+1,-NC 
EMID,ADD 
 
C*** 
C*** CAMBIA GLI ATTRIBUTI AD ALCUNI ELEMENTI 
C*** 
NSEL,,LOC,Y,-90+ANG2-0.001,-90+ANG1+0.001 
ESLN,,1 
EMODIF,ALL,REAL,2 
NSEL,,LOC,Y,-90-0.001,-90+ANG2+0.001 
ESLN,,1 
EMODIF,ALL,REAL,3 
ALLS 
C*** 
C*** CARICHI 
C*** 
SFGRAD,PRES,1,Y,90,PMX/90 
NSEL,,LOC,Y,-0.001,90.001 
ESLN,,1 
SFE,ALL,1,PRES,,PMX 
ALLS 
C*** 
C*** MATERIALE 
C***  
MP,EX,1,38000 
MP,EY,1,9000 
MP,EZ,1,9000 
MP,PRXY,1,0.3 
MP,PRYZ,1,0.3 
MP,PRXZ,1,0.3 
MP,GXY,1,3300 
MP,GXZ,1,3300 
MP,GYZ,1,3300 
 
TB,FAIL,1,1 
TBTEMP,,CRIT 
TBDATA,3,1 
TBTEMP,20 
TBDATA,10,SRPT,,SROT,,SROT, 
TBDATA,16,SRT 
C*** 
C*** VINCOLI 
C*** 
CSYS,0 
NSEL,,LOC,X,-0.1,0.001 
DSYMM,SYMM,X 
NSEL,,LOC,Z,-0.1,0.001 
D,ALL,ALL 
ALLS 
 
FINISH 
/SOLU 
SOLVE 
FINISH 
/POST1 
C*** 
C*** OPZIONI DI RAPPRESENTAZIONE 
C*** 
C*** scelta oggetti da rappresentare 
/PBC,ALL,,1 
/PBC,NFOR,,0 
/PBC,NMOM,,0 
/PBC,RFOR,,0 



/PBC,PATH,,0 
/PNUM,NODE,0 
/PSF,PRESS,NORM,2 
ALLS 
C*** scelta punto di vista 
/VIEW,,-0.3,0.2,-1 
C*** scelta disposizione finestra/legenda 
/PLOPTS,INFO,2 
C*** scelta colori 
/COLOR,PBAK,OFF 
C*** struttura con vincoli e carichi 
/TITLE,MODELLO 
ESEL,ALL 
EPLOT 
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0 
C*** deformata 
/PBC,ALL,,0 
/PSF,PRESS,NORM,0 
/PNUM,NODE,0 
/TITLE,DEFORMATA 
/VIEW,,0.3,0.3,-0.5 
PLDISP,1 
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0 
C*** caratteristiche di sollecitazione 
ETABLE,TSAIWU,NMISC,2         ! estrae il dato "tensione equivalente di 

Tsai-Wu" dal data base 
/TITLE,PARAMETRO DI DANNEGGIAMENTO - TASI-WU 
/GLINE,,-1 
PLETABLE, TSAIWU,AVG               ! rappresenta la forza normale 
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0 


